
1. Overview

This document lists a TTCN change needed to correct issues in the ATS ‘iwd-EUTRA-B2012-03_D13wk08’.

3. Corrections

Change 1 – Correction of RV value for SI3, SI4 and SI5 in TDD
	Function name
	f_EUTRA_SS_SetupSchedulingInfo

	Reason for change
	Acc. to TS 36.321 clause 5.3.1, the redundancy version of the received DL assignment is determined as per the calculation marked in red color below:

“ When the UE needs to read BCCH, the UE may, based on the scheduling information from RRC:

-
if a downlink assignment for this TTI has been received on the PDCCH for the SI-RNTI;

-
if the redundancy version is not defined in the PDCCH format:

-
the redundancy version of the received downlink assignment for this TTI is determined by RVK = ceiling(3/2*k) modulo 4, where k depends on the type of system information message: for SystemInformationBlockType1 message, k = (SFN/2) modulo 4, where SFN is the system frame number; for SystemInformation messages, k=i modulo 4, i =0,1,…, nsw–1, where i denotes the subframe number within the SI window nsw;

-
indicate a downlink assignment and redundancy version for the dedicated broadcast HARQ process to the HARQ entity for this TTI. “
Together with TS 36.523-3 Table 7.7.2-4, which specifies the System Information scheduling for TDD mode, and Table 7.7.2-2, which specifies the subframe Offset for TDD mode, SI3, SI4 and SI5 are scheduled in subframe i= 15 which leads acc. to the calculation above to k = 3 [k = 15 modulo 4] and thus to RV = 1 [RV = ceiling (1,5 * 3) modulo 4 = 5 modulo 4]. 
However, the function which takes care of Scheduling Info setup for the SS sets RV = 2, which is incorrect.


	Summary of change
	Changed setting of RV value for SI3, SI4 and SI5 in TDD mode to RV := 1.

	TTCN module
	CommonEUTRA\EUTRA_ConfigurationSteps.ttcn

	MCC160 Comment
	


Before change

…
  /*
   * @desc      extract scheduling info from SIB1; => whenever SIB scheduling is setup or changed
   *            this function returns required SS configuration
   * @param     p_SIB1
   * @param     p_FDD_TDD
   * @return    template (value) AllSiSchedul_Type
   * @status    APPROVED
   */
  function f_EUTRA_SS_SetupSchedulingInfo(SystemInformationBlockType1 p_SIB1,
                                          EUTRA_FDD_TDD_Mode_Type  p_FDD_TDD)
    return template (value) AllSiSchedul_Type
  {
    var template (value) SiSchedulList_Type v_SiSchedulList;
    var template (omit) SiSchedulList_Type v_SiSchedulSegmentList := omit; // @sic R5s110341 sic@
    var integer v_NoOfSIs := lengthof(p_SIB1.schedulingInfoList);     // TTCN-3 3.4.1: sizeof replaced by lengthof
    var integer v_SubframeOffset;                          /* offset within the SI-window */
    var integer v_RV;                                      /* @sic R5s100302 change 9; R5s110295 sic@ */
    var integer v_SI_I:=0;                                 /* @sic R5s110341 sic@ */
    var integer v_Segment_I := 0;                          /* @sic R5-112161 sic@ */
    var integer I;
    for (I := 0; I < v_NoOfSIs; I:= I + 1) {
      /* Assigne the subframe offset according to 36.523-3 Table 7.7.2.2;
         mapping of the SIBs to the SIs is specified in 36.508 cl. 4.4.3.1.2 "Scheduling of system information blocks"
         => there is a maximum of 4 SIs */
     select (p_FDD_TDD) {
       case (FDD) {
         select (I) {
           case (0) {   // Subframe offset for SI1
             v_SubframeOffset := 1;
             v_RV := 2;
           }
           case (1) { // Subframe offset for SI2
             v_SubframeOffset := 1;
             v_RV := 2;
           }
           // @sic R5-096691: 3 -> 1 sic@
           case (2) { // Subframe offset for SI3
             v_SubframeOffset := 3;
             v_RV := 1;                /* @sic R5s100302 change 9 sic@ */
           }
           case (3) {   // Subframe offset for SI4
             v_SubframeOffset := 7;
             v_RV := 1;                /* @sic R5s100302 change 9 R5s110016 sic@ */
           }
           case (4) {   // Subframe offset for SI5
             v_SubframeOffset := 7;
             v_RV := 1;                /* @sic R5-110471 sic@ */
           }
           case else {
             FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI");
           }
         }
       }
       case (TDD) {  // @sic R5-096691 sic@
         select (I) {
           case (0) { // Subframe offset for SI1
             v_SubframeOffset := 0;
             v_RV := 0;    /* @sic R5s110016 sic@ */
           }
           case (1) { // Subframe offset for SI2
             v_SubframeOffset := 0;
             v_RV := 0;    /* @sic R5s110016 sic@ */
           }
           case (2) { // Subframe offset for SI3
             v_SubframeOffset := 15;
             v_RV := 2;
           }
           case (3) {
             v_SubframeOffset := 15;
             v_RV := 2;
           }  // Subframe offset for SI4
           case (4) {
             v_SubframeOffset := 15;    // @sic r5-110471 sic@
             v_RV := 2;
           }  // Subframe offset for SI5
           case else {
             FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI");
           }
         }
       }
     }
     // For SibType11 and SibType12: add the sheduling information to the segmented sheduling List: v_SiSchedulSegmentList
     // For other SibTypes add the sheduling in the scheduling list v_SiSchedulList
     // @sic R5-112161 sic@
     if (fl_SibMappingInfo_Check(p_SIB1.schedulingInfoList[I].sib_MappingInfo, sibType11) or
         fl_SibMappingInfo_Check(p_SIB1.schedulingInfoList[I].sib_MappingInfo, sibType12_v920))  // @sic R5s110342 sic@
       {
         v_SiSchedulSegmentList[v_Segment_I] := cs_SiSchedul_Def(p_SIB1.schedulingInfoList[I].si_Periodicity, v_SubframeOffset);
         /* @sic R5s100302 change 9 sic@ */
         if (match(dci_1C, v_SiSchedulSegmentList[v_Segment_I].Window[0].DciInfo.Format)) {  // @sic R5s110341 sic@
           v_SiSchedulSegmentList[v_Segment_I].Window[0].DciInfo.RedundancyVersionList[0] := v_RV;
         }
         v_Segment_I := v_Segment_I+1;
       }
     else
       {
         v_SiSchedulList[v_SI_I] := cs_SiSchedul_Def(p_SIB1.schedulingInfoList[I].si_Periodicity, v_SubframeOffset);
         /* @sic R5s100302 change 9 sic@ */
         if (match(dci_1C, v_SiSchedulList[v_SI_I].Window[0].DciInfo.Format)) {
           v_SiSchedulList[v_SI_I].Window[0].DciInfo.RedundancyVersionList[0] := v_RV;
         }
         v_SI_I := v_SI_I+1;
       }
    }
    return cs_SiSchedulList_Def(p_SIB1.si_WindowLength, v_SiSchedulList, v_SiSchedulSegmentList);
  }
…
After change

…
  /*
   * @desc      extract scheduling info from SIB1; => whenever SIB scheduling is setup or changed
   *            this function returns required SS configuration
   * @param     p_SIB1
   * @param     p_FDD_TDD
   * @return    template (value) AllSiSchedul_Type
   * @status    APPROVED
   */
  function f_EUTRA_SS_SetupSchedulingInfo(SystemInformationBlockType1 p_SIB1,
                                          EUTRA_FDD_TDD_Mode_Type  p_FDD_TDD)
    return template (value) AllSiSchedul_Type
  {
    var template (value) SiSchedulList_Type v_SiSchedulList;
    var template (omit) SiSchedulList_Type v_SiSchedulSegmentList := omit; // @sic R5s110341 sic@
    var integer v_NoOfSIs := lengthof(p_SIB1.schedulingInfoList);     // TTCN-3 3.4.1: sizeof replaced by lengthof
    var integer v_SubframeOffset;                          /* offset within the SI-window */
    var integer v_RV;                                      /* @sic R5s100302 change 9; R5s110295 sic@ */
    var integer v_SI_I:=0;                                 /* @sic R5s110341 sic@ */
    var integer v_Segment_I := 0;                          /* @sic R5-112161 sic@ */
    var integer I;
    for (I := 0; I < v_NoOfSIs; I:= I + 1) {
      /* Assigne the subframe offset according to 36.523-3 Table 7.7.2.2;
         mapping of the SIBs to the SIs is specified in 36.508 cl. 4.4.3.1.2 "Scheduling of system information blocks"
         => there is a maximum of 4 SIs */
     select (p_FDD_TDD) {
       case (FDD) {
         select (I) {
           case (0) {   // Subframe offset for SI1
             v_SubframeOffset := 1;
             v_RV := 2;
           }
           case (1) { // Subframe offset for SI2
             v_SubframeOffset := 1;
             v_RV := 2;
           }
           // @sic R5-096691: 3 -> 1 sic@
           case (2) { // Subframe offset for SI3
             v_SubframeOffset := 3;
             v_RV := 1;                /* @sic R5s100302 change 9 sic@ */
           }
           case (3) {   // Subframe offset for SI4
             v_SubframeOffset := 7;
             v_RV := 1;                /* @sic R5s100302 change 9 R5s110016 sic@ */
           }
           case (4) {   // Subframe offset for SI5
             v_SubframeOffset := 7;
             v_RV := 1;                /* @sic R5-110471 sic@ */
           }
           case else {
             FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI");
           }
         }
       }
       case (TDD) {  // @sic R5-096691 sic@
         select (I) {
           case (0) { // Subframe offset for SI1
             v_SubframeOffset := 0;
             v_RV := 0;    /* @sic R5s110016 sic@ */
           }
           case (1) { // Subframe offset for SI2
             v_SubframeOffset := 0;
             v_RV := 0;    /* @sic R5s110016 sic@ */
           }
           case (2) { // Subframe offset for SI3
             v_SubframeOffset := 15;
             v_RV := 1;
           }
           case (3) {
             v_SubframeOffset := 15;
             v_RV := 1;
           }  // Subframe offset for SI4
           case (4) {
             v_SubframeOffset := 15;    // @sic r5-110471 sic@
             v_RV := 1;
           }  // Subframe offset for SI5
           case else {
             FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI");
           }
         }
       }
     }
     // For SibType11 and SibType12: add the sheduling information to the segmented sheduling List: v_SiSchedulSegmentList
     // For other SibTypes add the sheduling in the scheduling list v_SiSchedulList
     // @sic R5-112161 sic@
     if (fl_SibMappingInfo_Check(p_SIB1.schedulingInfoList[I].sib_MappingInfo, sibType11) or
         fl_SibMappingInfo_Check(p_SIB1.schedulingInfoList[I].sib_MappingInfo, sibType12_v920))  // @sic R5s110342 sic@
       {
         v_SiSchedulSegmentList[v_Segment_I] := cs_SiSchedul_Def(p_SIB1.schedulingInfoList[I].si_Periodicity, v_SubframeOffset);
         /* @sic R5s100302 change 9 sic@ */
         if (match(dci_1C, v_SiSchedulSegmentList[v_Segment_I].Window[0].DciInfo.Format)) {  // @sic R5s110341 sic@
           v_SiSchedulSegmentList[v_Segment_I].Window[0].DciInfo.RedundancyVersionList[0] := v_RV;
         }
         v_Segment_I := v_Segment_I+1;
       }
     else
       {
         v_SiSchedulList[v_SI_I] := cs_SiSchedul_Def(p_SIB1.schedulingInfoList[I].si_Periodicity, v_SubframeOffset);
         /* @sic R5s100302 change 9 sic@ */
         if (match(dci_1C, v_SiSchedulList[v_SI_I].Window[0].DciInfo.Format)) {
           v_SiSchedulList[v_SI_I].Window[0].DciInfo.RedundancyVersionList[0] := v_RV;
         }
         v_SI_I := v_SI_I+1;
       }
    }
    return cs_SiSchedulList_Def(p_SIB1.si_WindowLength, v_SiSchedulList, v_SiSchedulSegmentList);
  }
…  

