
1.1 Change 1

	Testcase name
	TC 9.2.3.3.5a

	Protocol
	EMM

	Function Name
	function f_GERAN_SwitchPowerOff

	Reason for change
	Inside "f_GERAN_SwitchPowerOff" function, once we receive CLASSMARKCHANGE message then, we are expecting UTRANCLASSMARKCHANGE if pc_UTRA is TRUE and then, GPRSSUSPENSIONREQUEST message. But, the last TWO messages can come in any order. Same is followed in the TTCN-2 scenarios.

	Summary of change
	Modified "f_GERAN_SwitchPowerOff" function, with interleave of GPRS Suspension Request and UtranClassMark Change messages so that they can be received in any order.

	Source of change
	GERAN_CommonFunctions.ttcn


Before:
	 function f_GERAN_SwitchPowerOff(GERAN_CellId_Type p_CellId,

                                                             UTRAN_GERAN_ATTACH_Type p_AttachType,

                                                             boolean p_USIM_Rmvd := false) runs on GERAN_PTC

  {

    var B2_Type v_Nmo := f_GERAN_NMO_Get ( p_CellId );

    var GERAN_POSTAMBLE_CONDITIONS v_Geran_Postamble_Condition := fl_GERAN_SetPostambleCondition (p_AttachType); // @sic R5s110176 sic@

    var TLLI v_TLLI := f_GERAN_TLLI_Get(p_CellId);

    timer t_Wait5s := 5.0;

    if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) // The UE can be switched off or USIM removed @sic R5s120597r1 sic@

      {

        f_UT_SwitchOffUE (UT,  p_USIM_Rmvd);

        if ((v_Geran_Postamble_Condition == GERAN_C3) or (v_Geran_Postamble_Condition == GERAN_C2 and v_Nmo == '01'B)) {

          // Then expect an IMSI Detach

          f_GERAN_RRConnEst(p_CellId); // @sic R5s120584 sic@

          G_L2.receive (car_G_L2_Estab_IND (p_CellId, ?, ?, ?, cr_G_Imsi_DetachInd));

          // @sic R5s120584, R5s120866 subject to prose CR approval sic@

          t_Wait5s.start;

          alt {

           [] G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_ClassmarkChangeAny))

             {

              t_Wait5s.stop;

              if (pc_UTRA) {

               G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_UtranClassmarkChange));

              }

              G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_GPRS_SuspensionReq(?, cr_RAI (omit))));

             }

           [] t_Wait5s.timeout {}

          }

        }

        if (v_Geran_Postamble_Condition != GERAN_C3) { // as long as GPRS is supported

          f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND ( cr_G_DetachRequest( cr_DetachTypeOnSwitchOff, cdr_MobileIdTMSI_tlv(?))));

        }

      }

    else

      {// The power must be removed from the UE

        f_UT_PowerOffUE (UT);

      }

    // @sic R5s120056 sic@

    if (tsc_TLLI_Initial != v_TLLI)  //if the TLLI hasn't changed, don't bother calling LLCAssignReq

      {

        f_LLC_TLLI_Assign (p_CellId, v_TLLI, tsc_TLLI_Initial);

        f_GERAN_OldTLLI_Set(p_CellId, v_TLLI);

        f_GERAN_TLLI_Set(p_CellId, tsc_TLLI_Initial);

      }

  }// end of f_GERAN_DetachOnSwitchOff


After:
	function f_GERAN_SwitchPowerOff(GERAN_CellId_Type p_CellId,

                                                             UTRAN_GERAN_ATTACH_Type p_AttachType,

                                                             boolean p_USIM_Rmvd := false) runs on GERAN_PTC

  {

    var B2_Type v_Nmo := f_GERAN_NMO_Get ( p_CellId );

    var GERAN_POSTAMBLE_CONDITIONS v_Geran_Postamble_Condition := fl_GERAN_SetPostambleCondition (p_AttachType); // @sic R5s110176 sic@

    var TLLI v_TLLI := f_GERAN_TLLI_Get(p_CellId);

    timer t_Wait5s := 5.0;

    if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) // The UE can be switched off or USIM removed @sic R5s120597r1 sic@

      {

        f_UT_SwitchOffUE (UT,  p_USIM_Rmvd);

        if ((v_Geran_Postamble_Condition == GERAN_C3) or (v_Geran_Postamble_Condition == GERAN_C2 and v_Nmo == '01'B)) {

          // Then expect an IMSI Detach

          f_GERAN_RRConnEst(p_CellId); // @sic R5s120584 sic@

          G_L2.receive (car_G_L2_Estab_IND (p_CellId, ?, ?, ?, cr_G_Imsi_DetachInd));

          // @sic R5s120584, R5s120866 subject to prose CR approval sic@

          t_Wait5s.start;

          alt {

           [] G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_ClassmarkChangeAny))

             {

              t_Wait5s.stop;

              if (pc_UTRA) {

              interleave { //@sic R5100744 sic@

            [] G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_UtranClassmarkChange)) {}

            [] G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_GPRS_SuspensionReq(?, cr_RAI (omit))));  {} /* @sic R5s120470 sic@ */

              }
              }

              else

              {

              G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_GPRS_SuspensionReq(?, cr_RAI (omit))));

              }

             }

           [] t_Wait5s.timeout {}

          }

        }

        if (v_Geran_Postamble_Condition != GERAN_C3) { // as long as GPRS is supported

          f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND ( cr_G_DetachRequest( cr_DetachTypeOnSwitchOff, cdr_MobileIdTMSI_tlv(?) ifpresent))); // @sic R5s120889 sic@

        }

      }

    else

      {// The power must be removed from the UE

        f_UT_PowerOffUE (UT);

      }

    // @sic R5s120056 sic@

    if (tsc_TLLI_Initial != v_TLLI)  //if the TLLI hasn't changed, don't bother calling LLCAssignReq

      {

        f_LLC_TLLI_Assign (p_CellId, v_TLLI, tsc_TLLI_Initial);

        f_GERAN_OldTLLI_Set(p_CellId, v_TLLI);

        f_GERAN_TLLI_Set(p_CellId, tsc_TLLI_Initial);

      }

  }// end of f_GERAN_DetachOnSwitchOff

